

More on User Defined Formats

1. Storing User Defined Formats in a Permanent Library

To define a permanent storage location for user defined formats, use the library= option on the PROC FORMAT statement to identify a permanent SAS data library (replace *SAS-data-library* with the location of your library in the libname statement below):

```
libname orion 'SAS-data-library';

proc format library=ORION;
    value $ctryfmt 'AU'='Australia' 'US'='United States'
        other='Miscoded';
    value tiers low-<50000 ='Tier 1' 50000-100000='Tier 2'
        100000<-high='Tier 3';
run;
```

The \$ctryfmt. and tiers. formats will be stored in a catalog in the ORION library.

When you reference a user-defined format, SAS will by default search first in the WORK library, and then in a library named LIBRARY, if it exists (you must assign this libref in order to use it). In order to identify additional libraries to search, and to define the search order, use the FMTSEARCH= SAS System option. **If your formats are stored in a library other than WORK or LIBRARY, you must issue this option in your SAS session prior to using the permanently saved formats.**

The WORK and LIBRARY libraries are always searched first, unless they are listed in the FMTSEARCH specification.

Libraries to be searched are specified in parentheses and separated by blanks. Search order using the following specification is WORK, LIBRARY, and ORION:

```
options fmtsearch=(orion);
```

If you want to alter the order for the WORK and/or LIBRARY libraries, specify them explicitly in the FMTSEARCH= option. Search order using the following specification is ORION, WORK, and then LIBRARY:

```
options fmtsearch=(orion work library);
```

2. Creating user-defined formats from a dataset or look-up table

If your values and labels exist in a dataset, you can use it as input to PROC FORMAT rather than typing VALUE statements. For more information, see the CNTLIN= option in the documentation for PROC FORMAT.

3. Displaying existing user defined format definitions

Use the FMTLIB option on the PROC FORMAT statement to display the definition for stored formats. Specify the library containing the format catalog of interest using the LIBRARY= option. The following PROC FORMAT step will create a report showing the definition of formats stored in the ORION library:

```
proc format fmtlib library=orion;
run;
```